

	
3GPP TSG-RAN5 Meeting #2023-TTCN email	R5s230587
Online, , 9th Dec 2022 - 31st Dec 2023
	CR-Form-v12.2

	CHANGE REQUEST

	

	
	36.523-3
	CR
	4738
	rev
	2
	Current version:
	18.0.0
	

	

	For HELP on using this form: comprehensive instructions can be found at 
http://www.3gpp.org/Change-Requests.

	



	Proposed change affects:
	UICC apps
	
	ME
	
	Radio Access Network
	
	Core Network
	



	

	Title:	
	Correction to LTE RRC testcase 8.5.4.1 

	
	

	Source to WG:
	Anritsu Ltd

	Source to TSG:
	R5

	
	

	Work item code:
	TEI8_Test
	
	Date:
	2023-05-18

	
	
	
	
	

	Category:
	F
	
	Release:
	Rel-18

	
	Use one of the following categories:
F  (correction)
A  (mirror corresponding to a change in an earlier 													release)
B  (addition of feature), 
C  (functional modification of feature)
D  (editorial modification)
Detailed explanations of the above categories can
be found in 3GPP TR 21.900.
	Use one of the following releases:
Rel-8	(Release 8)
Rel-9	(Release 9)
Rel-10	(Release 10)
Rel-11	(Release 11)
…
Rel-16	(Release 16)
Rel-17	(Release 17)
Rel-18	(Release 18)
Rel-19	(Release 19)

	
	

	Reason for change:
	1. When segmented ueCapabilityInformation is received, it is necessary to firstly decode it as UL-DCCH message rather than decoding it as eutraCapabilityMessage.
2. Present definition of function f_EUTRA_DecodeULDcchMsg doesn’t return the decoded UL-DCCH message but it is necessary to return the decoded value at times.
3. Failing to include requestedFrequencyBands_r11 IE in the ueCapabilityEnquiry message causes UE to not perform segmentation even though the UE supports UL segmentation.

	
	

	Summary of change:
	1. Segmented ueCapabilityInformation is initially decoded as UL-DCCH message and then decoding of ueCapabilityRAT_Container is done.
2. Added return value for function f_EUTRA_DecodeULDcchMsg and move the function definition to the module ‘EUTRA_CapabilityFunctions’
3. Populate requestedFrequencyBands_r11 IE and include it in the ueCapabilityEnquiry message when UL segmentation is supported by the UE (pc_LTE_UL_Segmentation=TRUE)

	
	

	Consequences if not approved:
	A conformant Rel16 UE will fail the test case

	
	

	Clauses affected:
	8.5.4.1

	
	

	
	Y
	N
	
	

	Other specs
	
	X
	 Other core specifications	
	

	affected:
	
	X
	 Test specifications
	

	(show related CRs)
	
	X
	 O&M Specifications
	

	
	

	Other comments:
	

	
	

	This CR's revision history:
	



Page 1


[bookmark: _Toc58338508]Table of Contents
1	Overview	4
2	Corrections required	4
2.1	Change 1	4
2.2	Change 2	6
2.3	Change 3	7







1. [bookmark: _Toc122434485][bookmark: _Toc52895930]Overview
This document lists all the changes needed to correct issues in the ATS iwd-TTCN3-B2022-09_D23wk12 related to the title of this CR.
Contact:	Kalahasti, Narendra
	Narendra.Kalahasti@anritsu.com

1. [bookmark: _Toc122434488][bookmark: _Toc295288959][bookmark: _Toc54186268]Corrections required
2.1 Change 1
	Function name
	f_EUTRA_GetCaps

	Reason for change
	When segmented ueCapabilityInformation is received, it is necessary to firstly decode it as UL-DCCH message rather than decoding it as eutraCapabilityMessage.

	Summary of change
	Segmented ueCapabilityInformation is initially decoded as UL-DCCH message and then decoding of ueCapabilityRAT_Container is done.

	TTCN module
	EUTRA_CapabilityFunctions.ttcn

	MCC160 Comment
	



Before change:
function f_EUTRA_GetCaps(EUTRA_CellId_Type p_CellId,
                            RRC_TransactionIdentifier p_RRC_TI) runs on EUTRA_PTC return UE_EUTRA_Capability
  { // @sic R5-217903 sic@
    var SRB_COMMON_IND v_ReceivedAspForUeCapabilityInfo;
    var UE_EUTRA_Capability v_RxEutraCapability;
    var SRB_COMMON_IND v_Received_DedicatedMessageSegment;
    var octetstring v_Rx_SegmentOct := ''O;
    var boolean v_Completed_AllSegments := false;
    var ULDedicatedMessageSegment_r16_IEs.segmentNumber_r16 v_MessageSegmentRx := 0; //INTEGER (0..15)

    // Rx UE Capability Information in segments or in one container
    // IF pc_LTE_UL_Segmentation and UE's encoded UECapabilityInformation message >= Max PDCP SDU size then segments will be handled else one container
    alt {
      // EXCEPTION: Segment handling is repeated a maximum of 16 times or till rrc-MessageSegmentType-16 IE within the UlDedicatedMessageSegment message indicate rrc-MessageSegmentType-r16 = lastSegment.
      [pc_LTE_UL_Segmentation] SRB.receive( car_SRB1_RrcPdu_IND(p_CellId, cr_508_ULDedicatedMessageSegment_r16(cr_ULDedicatedMessageSegment_r16_enabled(v_MessageSegmentRx))) ) -> value v_Received_DedicatedMessageSegment
      {
        v_Rx_SegmentOct := v_Rx_SegmentOct & v_Received_DedicatedMessageSegment.Signalling.Rrc.Dcch.message_.messageClassExtension.c2.ulDedicatedMessageSegment_r16.criticalExtensions.ulDedicatedMessageSegment_r16.rrc_MessageSegmentContainer_r16;   //compact segment
        if (v_Received_DedicatedMessageSegment.Signalling.Rrc.Dcch.message_.messageClassExtension.c2.ulDedicatedMessageSegment_r16.criticalExtensions.ulDedicatedMessageSegment_r16.rrc_MessageSegmentType_r16 == lastSegment){ // no more segments to receive
          v_Completed_AllSegments := true;
        }
        else { // notLastSegment received
          if (v_MessageSegmentRx == 15) { f_EUTRA_SetVerdictFailOrInconc(__FILE__,__LINE__, "Step 2a1: Received a notLastSegment as 16th segment from the UE");}
          v_MessageSegmentRx := v_MessageSegmentRx + 1;
          repeat;
        }
      }
      [not pc_LTE_UL_Segmentation] SRB.receive(car_SRB1_RrcPdu_IND(p_CellId, cr_508_ULDedicatedMessageSegment_r16(?))) // VerdictFail case
      {
        f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, "pc_LTE_UL_Segmentation not set for segmentation when UECapabilityInformation message >= Max PDCP SDU size");
      }
      // The UE transmits a UECapabilityInformation message including UE radio access capability information as per the ue-CapabilityRequest variable
      [] SRB.receive(car_SRB1_RrcPdu_IND(p_CellId, cr_508_UeCapabilityInformation(p_RRC_TI, cr_EutraCapInfo))) -> value v_ReceivedAspForUeCapabilityInfo
      {
        if (v_MessageSegmentRx>0) {
          f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, "UE radio access capability information reived after a segment"); // VerdictFail case: step 2b1 is mutually exclusive with 2ax
        }
      }
    };//end alt
        
    // Decode and store the UE-CapabilityRAT-Container
    if (v_Completed_AllSegments){ //UECapabilityInformation Rx. in segments
      if (lengthof(v_Rx_SegmentOct) < tsc_Max_LTE_PDCP_SDU_size) {//Set to <8188 by default - @sic R5s220330 sic@
        f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, "UE segmented the UECapabilityInformation message being smaller than Max PDCP SDU size");
      }
      v_RxEutraCapability := f_EUTRA_DecodeEutraCapMsg(v_Rx_SegmentOct);
      if (match (v_RxEutraCapability.accessStratumRelease, (rel8, rel9, rel10, rel11, rel12, rel13, rel14, rel15))) { // accessStratumRelease < rel16
        f_EUTRA_SetVerdictFailOrInconc (__FILE__, __LINE__, "AccessStratumRelease < release16 so segmentation is not allowed");
      }
    }
    else{ //UECapabilityInformation Rx. in one message
      v_Rx_SegmentOct  := v_ReceivedAspForUeCapabilityInfo.Signalling.Rrc.Dcch.message_.c1.ueCapabilityInformation.criticalExtensions.c1.ueCapabilityInformation_r8.ue_CapabilityRAT_ContainerList[0].ueCapabilityRAT_Container;
      if (lengthof(v_Rx_SegmentOct) > tsc_Max_LTE_PDCP_SDU_size){
        f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, "UE sent one segment of size > tsc_Max_LTE_PDCP_SDU_size");
      }
      v_RxEutraCapability := f_EUTRA_DecodeEutraCapMsg(v_ReceivedAspForUeCapabilityInfo.Signalling.Rrc.Dcch.message_.c1.ueCapabilityInformation.criticalExtensions.c1.ueCapabilityInformation_r8.ue_CapabilityRAT_ContainerList[0].ueCapabilityRAT_Container);
    }
        
    return v_RxEutraCapability;
  }


After change:

function f_EUTRA_GetCaps(EUTRA_CellId_Type p_CellId,
                            RRC_TransactionIdentifier p_RRC_TI) runs on EUTRA_PTC return UE_EUTRA_Capability
  { // @sic R5-217903 sic@
    var SRB_COMMON_IND v_ReceivedAspForUeCapabilityInfo;
    var UE_EUTRA_Capability v_RxEutraCapability;
	 var UL_DCCH_Message v_RxSegEutraCapability;
    var SRB_COMMON_IND v_Received_DedicatedMessageSegment;
    var octetstring v_Rx_SegmentOct := ''O;
    var boolean v_Completed_AllSegments := false;
    var ULDedicatedMessageSegment_r16_IEs.segmentNumber_r16 v_MessageSegmentRx := 0; //INTEGER (0..15)

    // Rx UE Capability Information in segments or in one container
    // IF pc_LTE_UL_Segmentation and UE's encoded UECapabilityInformation message >= Max PDCP SDU size then segments will be handled else one container
    alt {
      // EXCEPTION: Segment handling is repeated a maximum of 16 times or till rrc-MessageSegmentType-16 IE within the UlDedicatedMessageSegment message indicate rrc-MessageSegmentType-r16 = lastSegment.
      [pc_LTE_UL_Segmentation] SRB.receive( car_SRB1_RrcPdu_IND(p_CellId, cr_508_ULDedicatedMessageSegment_r16(cr_ULDedicatedMessageSegment_r16_enabled(v_MessageSegmentRx))) ) -> value v_Received_DedicatedMessageSegment
      {
        v_Rx_SegmentOct := v_Rx_SegmentOct & v_Received_DedicatedMessageSegment.Signalling.Rrc.Dcch.message_.messageClassExtension.c2.ulDedicatedMessageSegment_r16.criticalExtensions.ulDedicatedMessageSegment_r16.rrc_MessageSegmentContainer_r16;   //compact segment
        if (v_Received_DedicatedMessageSegment.Signalling.Rrc.Dcch.message_.messageClassExtension.c2.ulDedicatedMessageSegment_r16.criticalExtensions.ulDedicatedMessageSegment_r16.rrc_MessageSegmentType_r16 == lastSegment){ // no more segments to receive
          v_Completed_AllSegments := true;
        }
        else { // notLastSegment received
          if (v_MessageSegmentRx == 15) { f_EUTRA_SetVerdictFailOrInconc(__FILE__,__LINE__, "Step 2a1: Received a notLastSegment as 16th segment from the UE");}
          v_MessageSegmentRx := v_MessageSegmentRx + 1;
          repeat;
        }
      }
      [not pc_LTE_UL_Segmentation] SRB.receive(car_SRB1_RrcPdu_IND(p_CellId, cr_508_ULDedicatedMessageSegment_r16(?))) // VerdictFail case
      {
        f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, "pc_LTE_UL_Segmentation not set for segmentation when UECapabilityInformation message >= Max PDCP SDU size");
      }
      // The UE transmits a UECapabilityInformation message including UE radio access capability information as per the ue-CapabilityRequest variable
      [] SRB.receive(car_SRB1_RrcPdu_IND(p_CellId, cr_508_UeCapabilityInformation(p_RRC_TI, cr_EutraCapInfo))) -> value v_ReceivedAspForUeCapabilityInfo
      {
        if (v_MessageSegmentRx>0) {
          f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, "UE radio access capability information reived after a segment"); // VerdictFail case: step 2b1 is mutually exclusive with 2ax
        }
      }
    };//end alt
        
    // Decode and store the UE-CapabilityRAT-Container
    if (v_Completed_AllSegments){ //UECapabilityInformation Rx. in segments
      if (lengthof(v_Rx_SegmentOct) < tsc_Max_LTE_PDCP_SDU_size) {//Set to <8188 by default - @sic R5s220330 sic@
        f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, "UE segmented the UECapabilityInformation message being smaller than Max PDCP SDU size");
      }
      v_RxSegEutraCapability := f_EUTRA_DecodeULDcchMsg(v_Rx_SegmentOct);
      v_RxEutraCapability := f_EUTRA_DecodeEutraCapMsg(v_RxSegEutraCapability.message_.c1.ueCapabilityInformation.criticalExtensions.c1.ueCapabilityInformation_r8.ue_CapabilityRAT_ContainerList[0].ueCapabilityRAT_Container);
      if (match (v_RxEutraCapability.accessStratumRelease, (rel8, rel9, rel10, rel11, rel12, rel13, rel14, rel15))) { // accessStratumRelease < rel16
        f_EUTRA_SetVerdictFailOrInconc (__FILE__, __LINE__, "AccessStratumRelease < release16 so segmentation is not allowed");
      }
    }
    else{ //UECapabilityInformation Rx. in one message
      v_Rx_SegmentOct  := v_ReceivedAspForUeCapabilityInfo.Signalling.Rrc.Dcch.message_.c1.ueCapabilityInformation.criticalExtensions.c1.ueCapabilityInformation_r8.ue_CapabilityRAT_ContainerList[0].ueCapabilityRAT_Container;
      if (lengthof(v_Rx_SegmentOct) > tsc_Max_LTE_PDCP_SDU_size){
        f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, "UE sent one segment of size > tsc_Max_LTE_PDCP_SDU_size");
      }
      v_RxEutraCapability := f_EUTRA_DecodeEutraCapMsg(v_ReceivedAspForUeCapabilityInfo.Signalling.Rrc.Dcch.message_.c1.ueCapabilityInformation.criticalExtensions.c1.ueCapabilityInformation_r8.ue_CapabilityRAT_ContainerList[0].ueCapabilityRAT_Container);
    }
        
    return v_RxEutraCapability;
  }

2.2 Change 2
	Function name
	f_EUTRA_DecodeULDcchMsg

	Reason for change
	Present definition of function f_EUTRA_DecodeULDcchMsg doesn’t return the decoded UL-DCCH message but it is necessary to return the decoded value at times.

	Summary of change
	Added return value for function f_EUTRA_DecodeULDcchMsg and move the function definition to the module ‘EUTRA_CapabilityFunctions’ to be accessible by other testcases in LTE suite.

	TTCN module
	RRC_Others_PRO.ttcn

	MCC160 Comment
	



Before change:
function f_EUTRA_DecodeULDcchMsg(octetstring p_Octetstring)
  {
    var UL_DCCH_Message v_UL_Dcch_Msg;
    var bitstring v_Bitstring := oct2bit(p_Octetstring);
    var integer v_Result := decvalue(v_Bitstring, v_UL_Dcch_Msg);
    if (v_Result != 0) {
      FatalError(__FILE__, __LINE__, "UE EUTRA Capability cannot be decoded");
    }
  }


After change:

function f_EUTRA_DecodeULDcchMsg(octetstring p_Octetstring) return UL_DCCH_Message
  {
    var UL_DCCH_Message v_UL_Dcch_Msg;
    var bitstring v_Bitstring := oct2bit(p_Octetstring);
    var integer v_Result := decvalue(v_Bitstring, v_UL_Dcch_Msg);
    if (v_Result != 0) {
      FatalError(__FILE__, __LINE__, "UE EUTRA Capability cannot be decoded");
    }
	return v_UL_Dcch_Msg;
  }
2.3 Change 3
	Function name
	fl_EUTRA_FullCapabilityCheck

	Reason for change
	Failing to include requestedFrequencyBands_r11 IE in the ueCapabilityEnquiry message causes UE to not perform segmentation even though the UE supports UL segmentation.

	Summary of change
	Populate requestedFrequencyBands_r11 IE and include it in the ueCapabilityEnquiry message when UL segmentation is supported by the UE (pc_LTE_UL_Segmentation=TRUE)

	TTCN module
	LTE\8_5\RRC_Others.ttcn

	MCC160 Comment
	



Before change:
function fl_EUTRA_FullCapabilityCheck(EUTRA_CellId_Type p_CellId,
                                        RRC_TransactionIdentifier p_RRC_TI) runs on EUTRA_PTC
  {
    var UE_EUTRA_Capability v_ReceivedCapabilityMsg;
    var template UE_EUTRA_Capability v_EutraCapability;
    var template UE_EUTRA_Capability_v9a0_IEs v_Expected_UE_EUTRA_Capability_v9a0_IEs := omit;
    
    // Transmit UE Capability Enquiry message to ask for EUTRA Capability information
    SRB.send(cas_SRB1_RrcPdu_REQ(p_CellId, cs_TimingInfo_Now, cs_508_UeCapabilityEnquiry(p_RRC_TI,-,-,-,-,enabled))); // @sic R5-217903 sic@
    
    // Receive UE Capability Information message for EUTRA capabilities and decode
    // Either a container is received or segments from which the container is to be established (only Rel-16 onwards)
    v_ReceivedCapabilityMsg := f_EUTRA_GetCaps(p_CellId, p_RRC_TI);  // @sic R5-217903 sic@
    
    // Skipped code

  } // fl_EUTRA_FullCapabilityCheck


After change:

import from EUTRA_AuxiliaryCapCheckFunctions all; 
import from EUTRA_Parameters all;

function fl_EUTRA_FullCapabilityCheck(EUTRA_CellId_Type p_CellId,
                                        RRC_TransactionIdentifier p_RRC_TI) runs on EUTRA_PTC
  {
    var UE_EUTRA_Capability v_ReceivedCapabilityMsg;
    var template UE_EUTRA_Capability v_EutraCapability;
    var template UE_EUTRA_Capability_v9a0_IEs v_Expected_UE_EUTRA_Capability_v9a0_IEs := omit;
	 var UECapabilityEnquiry_v1180_IEs.requestedFrequencyBands_r11 v_RequestedFrequencyBands := {px_ePrimaryFrequencyBand}; 
	var integer v_Index := 0;
	var bitstring v_SuppEutraBandsAccToPICS := f_DeriveSuppEutraBandsFromPics();
	var UECapabilityEnquiry_v1180_IEs.requestedFrequencyBands_r11 v_BandToBeAdded;
	
	if(pc_LTE_UL_Segmentation) {
// Get bands to be requested, from all the supported bands
    do {
      if(v_SuppEutraBandsAccToPICS[v_Index]== '1'B and (v_Index+1) != px_ePrimaryFrequencyBand) { // if the UE supposts that band and it is not the primary band (which is already added in the list)
        v_BandToBeAdded := {v_Index+1};
        v_RequestedFrequencyBands := v_RequestedFrequencyBands & v_BandToBeAdded; // add supported band to the list
      }
      v_Index := v_Index + 1;
    } while ((lengthof(v_RequestedFrequencyBands) < 16) or ((v_Index < lengthof(v_SuppEutraBandsAccToPICS)) and (lengthof(v_SuppEutraBandsAccToPICS) < 16)));
    
    // Transmit UE Capability Enquiry message to ask for EUTRA Capability information
    SRB.send(cas_SRB1_RrcPdu_REQ(p_CellId, cs_TimingInfo_Now, cs_508_UeCapabilityEnquiry(p_RRC_TI,-,-,-,-,enabled ,v_RequestedFrequencyBands))); // @sic R5-217903 sic@
	 } else {
	 // Transmit UE Capability Enquiry message to ask for EUTRA Capability information
    SRB.send(cas_SRB1_RrcPdu_REQ(p_CellId, cs_TimingInfo_Now, cs_508_UeCapabilityEnquiry(p_RRC_TI,-,-,-,-,enabled ))); // @sic R5-217903 sic@
	 }
    
    // Receive UE Capability Information message for EUTRA capabilities and decode
    // Either a container is received or segments from which the container is to be established (only Rel-16 onwards)
    v_ReceivedCapabilityMsg := f_EUTRA_GetCaps(p_CellId, p_RRC_TI);  // @sic R5-217903 sic@
    
    // Skipped code

  } // fl_EUTRA_FullCapabilityCheck

